“Everything Is Everything” Revisited: Shapeshifting Data Types with Isomorphisms and Hylomorphisms

Paul Tarau
Department of Computer Science and Engineering
University of North Texas
Denton, TX 76203-6886, USA

This paper is an exploration of isomorphisms between elementary data types (e.g., natural numbers, sets, finite functions, graphs, hypergraphs) and their extension to hereditarily finite universes through hylomorphisms derived from ranking/unranking and pairing/unpairing operations. An embedded higher order combinator language provides any-to-any encodings automatically. A few examples of free algorithms obtained by transferring operations between data types are shown. Other applications range from stream iterators on combinatorial objects to succinct data representations and the generation of random instances.

1. Introduction

Kolmogorov-Chaitin algorithmic complexity is based on the existence of various equivalent representations of data objects, and in particular (minimal) programs that produce them in a given language and encoding [1-3]. Analogical/metaphorical thinking routinely shifts entities and operations from one field to another hoping to uncover similarities in representation or use [4]. Compilers convert programs from human-centered to machine-centered representations. Complexity classes are defined through compilation with limited resources (time or space) to similar problems [5, 6]. Mathematical theories often borrow proof patterns and reasoning techniques across close and some time not so close fields.

A relatively small number of universal data types are used as basic building blocks in programming languages and their runtime interpreters, corresponding to well-tested mathematical abstractions like sets, functions, graphs, groups, categories, and others.

Hence, everyone explicitly or implicitly knows that ultimately, “everything is everything” through lower common denominators like bitstring representations in computer memory. From hackers and combinatorialists and experimental mathematicians, it is not uncommon to shapeshift between various data types. Means as
simple as binary editors, union types, or overlapping variable definitions are generously providing such alternate views.

A less obvious leap is that if heterogeneous objects can be seen in some way as isomorphic, then we can share them and compress the underlying informational universe by collapsing isomorphic encodings of data or programs whenever possible.

The added benefit of these “shapeshifting” data types is that the functors transporting their data content will also transport their operations, resulting in shortcuts that provide, for free, implementations of interesting algorithms. The simplest instance is the case of isomorphisms, which are reversible mappings that also transport operations. Typically, such isomorphisms show up as encodings to some simpler and easier to manipulate representation, for instance, natural numbers, finite sets, or finite sequences. Such encodings can be traced back to Gödel numberings [7, 8] associated to formulas, but a wide diversity of common computer operations, ranging from wireless data transmissions to cryptographic codes, qualify.

Encodings between data types provide a variety of services, ranging from free iterators and random objects to data compression and succinct representations. Tasks like serialization and persistence are facilitated by simplification of reading or writing operations without the need of special purpose parsers. Sensitivity to internal data representation format or size limitations can be circumvented without extra programming effort.

In the context of algorithmic information theory, data structures can be interpreted like graphs and program constructs like loops or recursion as compression mechanisms that focus on sharing and reusing equivalent blocks of information. In this case, maximal sharing acts as the dual of minimal program+input size. With this in mind, shapeshifting through a uniform set of encodings would extend sharing opportunities across heterogeneous data and code types.

This paper is organized as follows. Section 2 introduces the general framework for the paper, in the form of an embedded data transformation language, that is applied in Section 3 to obtain encodings of some basic data types. Section 4 discusses a mechanism for lifting our transformations to hereditarily finite functions and sets. After reviewing some classic pairing functions, Section 5 introduces pairing/unpairing operations that are used in Section 7 to provide bijective encodings of digraphs, directed acyclic graphs (DAGs), and hypergraphs as natural numbers. Section 8 describes applications that focus on combinatorial generation, experimental mathematics, random instances, and succinct representations of various data types. Sections 9 and 10 discuss related work, future work, and conclusions.

Here are the main contributions of this paper (with section/subsection numbers in parenthesis):

- a general framework for bijective encodings between heterogeneous data types in Haskell and an embedded combinator language providing
automatic any-to-any encoding by routing through a common representation (2)

- efficient and simple encodings of finite sequences (3.2), digraphs, DAGs, and hypergraphs (7)
- a mechanism for lifting encodings to hereditarily finite data types (4) and its application to derive Ackermann’s encoding for hereditarily finite sets
- a new instance of hereditarily finite representations derived from finite function encodings (4.1.2)
- a bijective natural number encoding of list processing code (6)
- a presentation of our results as a self-contained literate Haskell program directly testable for technical correctness and reusable as a public domain Haskell library

## 2. An Embedded Data Transformation Language

We start by designing an embedded transformation language as a set of operations on a groupoid of isomorphisms. We then extend it with higher order combinators mediating the composition of encodings and the transfer of operations between data types.

### 2.1 The Groupoid of Isomorphisms

We implement an isomorphism between two objects \( X \) and \( Y \) as a Haskell data type encapsulating a bijection \( f \) and its inverse \( g \). We will call the `from` function the first component (known as a `section` in category theory parlance) and the `to` function the second component (a `retraction`) defining the isomorphism. We can organize isomorphisms as a `groupoid` as follows:

\[
X \xrightarrow{f = g^{-1}} Y.
\]

```haskell
data Iso a b = Iso (a -> b) (b -> a)

from (Iso f _) = f
to (Iso _ g) = g

compose :: Iso a b -> Iso b c -> Iso a c
compose (Iso f g) (Iso f' g') = Iso (f' . f) (g . g')

itself = Iso id id

invert :: Iso a b -> Iso b a
invert (Iso f g) = Iso g f
```

"Everything Is Everything” Revisited 477

Complex Systems, 18 © 2010 Complex Systems Publications, Inc.
https://doi.org/10.25088/ComplexSystems.18.4.475
Assuming that for any pair of type \( Iso \ a \ b \), \( f \circ g = id_a \) and \( g \circ f = id_b \), we can now formulate laws about isomorphisms that can be used to test the correctness of implementations with tools like QuickCheck [9].

**Proposition 1.** The data type \( Iso \) has a groupoid structure, that is, the compose operation, when it is defined, is associative, itself acts as an identity element, and \( invert \) computes the inverse of an isomorphism.

We can transport operations from one object to another with borrow and lend combinators defined as follows.

\[
\begin{align*}
\text{borrow} &:: Iso \ t \ s \to (t \to t) \to s \to s \\
\text{borrow} (Iso f g) h x &= f (h (g x)) \\
\text{borrow2} (Iso f g) h x y &= f (h (g x) (g y)) \\
\text{borrowN} (Iso f g) h x s &= f (h (\text{map} g x)) \\
\text{lend} &:: Iso \ s \ t \to (t \to t) \to s \to s \\
\text{lend} &= \text{borrow} \cdot \text{invert} \\
\text{lend2} &= \text{borrow2} \cdot \text{invert} \\
\text{lendN} &= \text{borrowN} \cdot \text{invert}
\end{align*}
\]

The combinators \( fit \) and \( retrofit \) just transport an object \( x \) through an isomorphism and apply to it an operation \( op \) available on the other side.

\[
\begin{align*}
\text{fit} &:: (b \to c) \to Iso \ a \ b \to a \to c \\
\text{fit} \ op \ iso \ x &= op ((\text{from} \ iso) x) \\
\text{retrofit} &:: (a \to c) \to Iso \ a \ b \to b \to c \\
\text{retrofit} \ op \ iso \ x &= op ((\text{to} \ iso) x)
\end{align*}
\]

We can see the combinators from, to, compose, itself, invert, borrow, lend, fit, and others as part of an embedded data transformation language. Note that in this design, we borrow from our strongly typed host programming language its abstraction layers and type safety mechanisms that continue to check the semantic validity of the embedded language constructs. For instance, the type checker ensures that two isomorphisms \( f \) and \( g \) can be composed if and only if the target of \( f \) is the same as the source of \( g \).

### 2.2 Choosing a Root

To avoid defining \( \frac{n(n-1)}{2} \) isomorphisms between \( n \) objects, we choose a Root object to/from which we will actually implement isomorphisms. We will extend our embedded combinator language using the groupoid structure of the isomorphisms to connect any two objects through isomorphisms to/from the Root.

Choosing a Root object is somewhat arbitrary, but it makes sense to pick a representation that is relatively easily convertible to various
others, efficiently implementable and, last but not least, scalable to accommodate large objects up to the runtime system’s actual memory limits.

We will choose as our Root object finite sequences of natural numbers. They can be seen as finite functions from an initial segment of $\mathbb{N}$, say $[0..n]$, to $\mathbb{N}$. We will represent them as lists, that is, their Haskell type is $[\text{Nat}]$. Alternatively, an array representation can be chosen.

```
type Nat = Integer
type Root = [Nat]
```

We can now define an Encoder as an isomorphism connecting an object to Root

```
type Encoder a = Iso a Root
```

together with the combinators with and as providing an embedded transformation language for routing isomorphisms through two encoders.

```
with :: Encoder a -> Encoder b -> Iso a b
with this that = compose this (invert that)
```

```
as :: Encoder a -> Encoder b -> b -> a
as that this = to (with that this)
```

The combinator with turns two encoders into an arbitrary isomorphism, that is, it acts as a connection hub between their domains. The combinator as adds a more convenient syntax, such that converters between $A$ and $B$ can be designed as in the following template.

```
a2b x = as B A x
b2a x = as A B x
```

We will provide extensive use cases for these combinators as we populate our groupoid of isomorphisms. Given that $[\text{Nat}]$ has been chosen as the root, we define our finite function data type fun simply as the identity isomorphism on sequences in $[\text{Nat}]$.

```
fun :: Encoder [Nat]
fun = itself
```

## 3. Extending the Groupoid of Isomorphisms

We now populate our groupoid of isomorphisms with combinators based on a few primitive converters.
### 3.1 An Isomorphism to Finite Sets of Natural Numbers

The isomorphism is specified with two bijections, `set2fun` and `fun2set`.

```haskell
set :: Encoder [Nat]
set = Iso set2fun fun2set
```

While finite sets and sequences share a common representation `[Nat]`, sets are subject to the implicit constraint that all their elements are distinct. Such constraints can be regarded as “laws” that we assume about a given data type when needed, restricting it to the appropriate domain of the underlying mathematical concept. This suggests that a set like `{7, 1, 4, 3}` could be represented by first ordering it as `{1, 3, 4, 7}` and then computing the differences between consecutive elements. This gives `[1, 2, 1, 3]`, with the first element 1 followed by the increments `[2, 1, 3]`. To turn the set into a bijection, including 0 as a possible member of a sequence, another adjustment is needed: elements in the sequence of increments should be replaced by their predecessors. This gives `[1, 1, 0, 2]` as implemented by `set2fun`.

```haskell
set2fun is | is_set is =
  map pred (genericTake l ys) where
  ns=sort is
  l=genericLength ns
  next n | n≥0 = succ n
  xs=(map next ns)
  ys=(zipWith (-) (xs++[0]) (0:xs))
  is_set ns = ns===nub ns
```

It can now be verified easily that incremental sums of the successors of numbers in such a sequence return the original set in sorted form, as implemented by `fun2set`.

```haskell
fun2set ns =
  map pred (tail (scanl (+) 0 (map next ns))) where
  next n | n≥0 = succ n
```

The resulting encoder `set` is now ready to interoperate with another encoder.

```haskell
*ISO> as set fun [0,1,0,0,4]
[0,2,3,4,9]
*ISO> as fun set [0,2,3,4,9]
[0,1,0,0,4]
```

As the example shows, this encoding maps arbitrary lists of natural numbers representing finite functions to strictly increasing sequences of (distinct) natural numbers representing sets.

### 3.2 Folding Sets into Natural Numbers

We can fold a set, represented as a list of distinct natural numbers, into a single natural number, reversibly, by observing that it can be

---

*Complex Systems, 18 © 2010 Complex Systems Publications, Inc.*

https://doi.org/10.25088/ComplexSystems.18.4.475
seen as the list of exponents of 2 in the number’s base 2 representation.

\[
\text{nat\_set} = \text{Iso nat2set set2nat}
\]

\[
\text{nat2set\_n} | n\geq0 = \text{nat2exp\_s n} 0 \text{ where}
\]

\[
\text{nat2exp\_s 0} _ = [ ]
\]

\[
\text{nat2exp\_s n} x =
\]

\[
\text{if (even n) then xs else (x:xs) where}
\]

\[
xs = \text{nat2exp\_s (n \div 2)} (\text{succ } x)
\]

\[
\text{set2nat\_ns} | \text{is\_set\_ns} = \text{sum (map (2^) ns)}
\]

We standardize this pair of operations as an Encoder for a natural number using our Root as a mediator.

\[
\text{nat} :: \text{Encoder Nat}
\]

\[
\text{nat} = \text{compose nat\_set set}
\]

The resulting encoder nat is now ready to interoperate with any other encoder.

*ISO> as fun nat 2008
[3,0,1,0,0,0,0]
*ISO> as set nat 2008
[3,4,6,7,8,9,10]
*ISO> as nat set [3,4,6,7,8,9,10] 2008
*ISO> lend nat reverse 2008
1135
*ISO> lend nat\_set reverse 2008
2008
*ISO> borrow nat\_set succ [1,2,3]
[0,1,2,3]
*ISO> as set nat 42
[1,3,5]
*ISO> fit length nat 42
3
*ISO> retrofit succ nat\_set [1,3,5]
43

The reader might notice at this point that we have already made a full circle—finite sets can be seen as instances of finite sequences. Injective functions that are not surjections with wider and wider gaps can be generated using the fact that one of the representations is information theoretically “denser” than the other, for a given range.

*ISO> as set fun [0,1,2,3]
[0,2,5,9]
*ISO> as set fun (as set fun [0,1,2,3])
[0,3,9,19]
*ISO> as set fun (as set fun (as set fun [0,1,2,3]))
[0,4,14,34]
4. Generic Unranking and Ranking Hylomorphisms

The ranking problem for a family of combinatorial objects is finding a unique natural number associated to it, called its rank. The inverse unranking problem consists of generating a unique combinatorial object associated to each natural number.

4.1 Hereditarily Finite Data Types

The unranking operation is seen here as an instance of a generic anamorphism mechanism (an unfold operation), while the ranking operation is seen as an instance of the corresponding catamorphism (a fold operation) [10, 11]. Together they form a mixed transformation called hylomorphism. We use such hylomorphisms to lift isomorphisms between lists and natural numbers to isomorphisms between a derived “self-similar” tree data type and natural numbers. In particular, we will derive Ackermann’s encoding from hereditarily finite sets to natural numbers.

The data type representing hereditarily finite structures will be a generic multiway tree with a single leaf type []:

\[
data T = H T \text{ deriving (Eq,Ord,Read,Show)}
\]

\[
type Ts = [T]
\]

The two sides of our hylomorphism are parameterized by two transformations \( f \) and \( g \) forming an isomorphism \( \text{Iso } f \ g \).

\[
\begin{align*}
\text{unrank} & : (a \rightarrow [a]) \rightarrow a \rightarrow T \\
\text{unranks} & : (a \rightarrow [a]) \rightarrow [a] \rightarrow Ts \\
\text{unrank } f \ n & = H (\text{unranks } f \ (f \ n)) \\
\text{unranks } f \ ns & = \text{map } (\text{unrank } f) \ ns \\
\text{rank} & : ([b] \rightarrow b) \rightarrow T \rightarrow b \\
\text{ranks} & : ([b] \rightarrow b) \rightarrow Ts \rightarrow [b] \\
\text{rank } g \ (H \ ts) & = g \ (\text{ranks } g \ ts) \\
\text{ranks } g \ ts & = \text{map } (\text{rank } g) \ ts
\end{align*}
\]

Both combinators can be seen as a form of “structured recursion” that propagate a simpler operation guided by the structure of the data type. For instance, the size of a tree of type \( T \) is obtained as:

\[
tsize = \text{rank } \lambda x s \rightarrow 1 + (\text{sum } x s)
\]

Note also that \text{unrank} and \text{rank} work on \( T \) in cooperation with \text{unranks} and \text{ranks} working on \( Ts \).

We can now combine an anamorphism+catamorphism pair into an isomorphism \( \text{hylo} \) defined with \text{rank} and \text{unrank} on the corresponding hereditarily finite data types.

\[
\begin{align*}
\text{hylo} & : \text{Iso } b \ [b] \rightarrow \text{Iso } T \ b \\
\text{hylo} \ (\text{Iso } f \ g) & = \text{Iso } (\text{rank } g) \ (\text{unrank } f) \\
\text{hylos} & : \text{Iso } b \ [b] \rightarrow \text{Iso } Ts \ [b] \\
\text{hylos} \ (\text{Iso } f \ g) & = \text{Iso } (\text{ranks } g) \ (\text{unranks } f)
\end{align*}
\]
4.1.1 Hereditarily Finite Sets

Hereditarily finite sets will be represented as an encoder for the tree type T.

\[
\text{hfs} :: \text{Encoder T}
\]
\[
\text{hfs} = \text{compose (hylo nat_set) nat}
\]

The \text{hfs} encoder can now borrow operations from sets or natural numbers as follows.

\[
\text{hfs\_union} = \text{borrow\_2 (with set hfs) union}
\]
\[
\text{hfs\_succ} = \text{borrow (with nat hfs) succ}
\]
\[
\text{hfs\_pred} = \text{borrow (with nat hfs) pred}
\]

*ISO> hfs\_succ (H [])
H [H []]

*ISO> hfs\_union (H [H []]) (H [])
H [H []]

Otherwise, hylomorphism induced isomorphisms work as usual with our embedded transformation language.

*ISO> as hfs nat 42
H [H [H [], H [H [], H [H []]], H [H [], H [H []]]]]

One can notice that we have just derived, as a “free algorithm”, Ackermann’s encoding from hereditarily finite sets to natural numbers:

\[
f(x) = \text{if } x = \emptyset \text{ then } 0 \text{ else } \sum_{a \in x} 2^{f(a)}
\]

together with its inverse:

\[
\text{ackermann} = \text{as nat hfs}
\]
\[
\text{inverse\_ackermann} = \text{as hfs nat}
\]

4.1.2 Hereditarily Finite Functions

The same tree data type can host a hylomorphism derived from finite functions instead of finite sets.

\[
\text{hff} :: \text{Encoder T}
\]
\[
\text{hff} = \text{compose (hylo nat) nat}
\]

The \text{hff} encoder can be seen as another free algorithm, providing data compression/succinct representation for hereditarily finite sets. Note, for instance, the significantly smaller tree size in the following.

*ISO> as hff nat 42
H [H [H [], H [H [], H [H []]]]]

As the cognoscenti might observe, this is explained by the fact that \text{hff} provides higher information density than \text{hfs} by incorporating order information that matters in the case of a sequence, and is ignored in the case of a set.
A pairing function is an isomorphism $f : \text{Nat} \times \text{Nat} \rightarrow \text{Nat}$. Its inverse is called unpairing. We now introduce an unusually simple pairing function (also mentioned in [12, p. 142]). The function bitpair works by splitting a number's big endian bitstring representation into odd and even bits, while its inverse bitunpair blends the odd and even bits back together.

```
source (x,_) = x
target (_ ,y) = y
```

```
bitpair :: Nat2 -> Nat
bitpair (i,j) =
  set2nat ((evens i) ++ (odds j)) where
  evens x = map (2*) (nat2set x)
  odds y = map succ (evens y)
```

```
bitunpair :: Nat -> Nat2
bitunpair n = (f xs, f ys) where
  (xs,ys) = partition even (nat2set n)
  f = set2nat . (map ('div' 2))
```

The transformation of the bitlists is shown in the following example with bitstrings aligned.

```
*ISO> bitunpair 200
  60 : [0, 0, 0, 1, 1, 0, 1, 1, 1, 1, 1]
  26 : [0, 0, 1, 1, 1, 1]
```

We can derive the following encoder:

```
nat2 :: Encoder Nat2
nat2 = compose (Iso bitpair bitunpair) nat
```

working as follows.

```
*ISO> as nat2 nat 200
(60,26)
```

Pairing/unpairing operations are important because they can encode complex objects (when interpreted as cons+head+tail operations) and ultimately code, for instance, simple LISP or \(\lambda\)-calculus programs and interpreters.
## 6. Cons-Lists with Pairing/Unpairing

The simplest application of pairing/unpairing operations is the encoding of cons-lists of natural numbers, defined as the CList data type.

```haskell
data CList = Atom Nat | Cons CList CList
    deriving (Eq,Ord,Show,Read)
```

First, to provide an infinite supply of atoms, we encode them as even numbers.

```haskell
to_atom n = 2*n
from_atom a | is_atom a = a \div\ 2
is_atom n = even n && n > 0
```

Next, because we want atoms and cons cells to be disjoint, we encode the latter as odd numbers.

```haskell
is_cons n = odd n && n > 0
decons z | is_cons z = bitunpair ((z-1) \div\ 2)
cons x y = 2*(bitpair (x,y))+1
```

We can deconstruct a natural number by recursing over applications of the unpairing-based decons combinator.

```haskell
nat2cons n | is_atom n = Atom (from_atom n)
nat2cons n | is_cons n = Cons (nat2cons hd)
    (nat2cons tl) where
    (hd,tl) = decons n
```

We can reverse this process by recursing with the cons combinator on the CList data type.

```haskell
cons2nat (Atom a) = to_atom a
cons2nat (Cons h t) = cons (cons2nat h) (cons2nat t)
```

The following example shows both transformations as inverses.

```haskell
*ISO> nat2cons 123456789
Cons
    (Atom 2512)
    (Cons
        (Cons
            (Cons
                (Cons (Atom 0) (Atom 0))
                (Cons (Atom 0) (Atom 0))
            )
            (Atom 1)
        )
        (Atom 27)
    )
*ISO> cons2nat it
123456789
```

We obtain the encoder:

```haskell
clist :: Encoder CList
clist = compose (Iso cons2nat nat2cons) nat
```
The encoder works as follows:

\[
\begin{align*}
\text{ISO} & \text{ as } \text{clist nat 101} \\
& \text{Cons (Atom 2) (Cons (Atom 0) (Cons (Atom 0) (Atom 0)))}
\end{align*}
\]

and can be used to generate random LISP-like data and code skeletons from natural numbers.

## 7. Directed Graphs and Hypergraphs

We now show that more complex data types, such as digraphs and hypergraphs, have extremely simple encoders. This emphasizes once more the importance of compositionality in the design of our embedded transformation language.

### 7.1 Encoding Directed Graphs

We can find a bijection from directed graphs (with no isolated vertices, corresponding to their view as binary relations) to finite sets by fusing their list of ordered pair representation into finite sets with a pairing function.

\[
\begin{align*}
digraph2set & : \{\text{Nat}2\} \rightarrow \{\text{Nat}\} \\
set2digraph & : \{\text{Nat}\} \rightarrow \{\text{Nat}2\}
\end{align*}
\]

\[
\begin{align*}
digraph2set \text{ ps} & = \text{map bitpair ps} \\
set2digraph \text{ ns} & = \text{map bitunpair ns}
\end{align*}
\]

Here is the resulting encoder:

\[
\begin{align*}
digraph : & \text{ Encoder } [\text{Nat}2] \\
digraph & = \text{compose (Iso digraph2set set2digraph) set}
\end{align*}
\]

working as follows:

\[
\begin{align*}
*\text{ISO}> & \text{ as digraph nat 2009} \\
& [(0,0),(1,1),(2,0),(2,1),(3,1),(0,2),(1,2),(0,3)] \\
*\text{ISO}> & \text{ as nat digraph it} \\
& 2009
\end{align*}
\]

### 7.2 Encoding Directed Acyclic Graphs

Encoders for DAGs can be derived from the encoding of digraphs under the assumption that they are canonically represented by pairs of edges such that the first element of the pair is strictly smaller.

After defining:

\[
\begin{align*}
digraph2dag & = \text{map } f \text{ where } f (x,y) = (x,y+x+1) \\
dag2digraph & = \text{map } f \text{ where } f (x,y) | y>x = (x,y-x-1)
\end{align*}
\]

we obtain the encoder:

\[
\begin{align*}
dag : & \text{ Encoder } [\text{Nat}2] \\
dag & = \text{compose (Iso dag2digraph digraph2dag ) digraph}
\end{align*}
\]
working as follows:

```haskell
*ISO> as nat dag
[(0,1),(1,2),(0,2),(1,3),(2,3),(3,4),(4,5),(1,6)]
```

Clearly, this mapping shows that there is a bijection between the set of finite digraphs and finite DAGs that preserves the number of edges. Note also that there is an obvious mapping between DAGs and unordered graphs, without self-loops, obtained by always directing edges from vertices with lower indices to vertices with strictly higher indices.

### 7.3 Encoding Hypergraphs

**Definition 1.** A hypergraph (also called set system) is a pair \( H = (X, E) \) where \( X \) is a set and \( E \) is a set of nonempty subsets of \( X \).

We can easily derive a bijective encoding of hypergraphs, represented as sets of nonempty sets.

```haskell
set2hypergraph :: [Nat] \rightarrow [[Nat]]
hypergraph2set :: [[Nat]] \rightarrow [Nat]
set2hypergraph = map (nat2set . succ)
hypergraph2set = map (pred . set2nat)
```

The resulting encoder is:

```haskell
hypergraph :: Encoder [[Nat]]
hypergraph = compose (Iso hypergraph2set set2hypergraph) set
```

working as follows:

```haskell
*ISO> as hypergraph nat 2009
[[0],[2],[0,2],[0,1,2],[3],[0,3],[1,3],[0,1,3]]
```

### 8. Applications

Besides their utility as a uniform basis for a general purpose data conversion library, let us point out some specific applications of our isomorphisms.
8.1 Combinatorial Generation

A free combinatorial generation algorithm (providing a constructive proof of recursive enumerability) for a given structure is obtained simply through an isomorphism from nat.

\[
\begin{align*}
\text{nth} & : \text{Encoder} \ a \rightarrow \text{Nat} \rightarrow \ a \\
\text{nth thing} & = \text{as thing nat} \\
\text{nths} & : \text{Encoder} \ a \rightarrow \text{[Nat]} \rightarrow \text{[a]} \\
\text{nths thing} & = \text{map (nth thing)} \\
\text{stream_of} & : \text{Encoder} \ a \rightarrow \text{[a]} \\
\text{stream_of thing} & = \text{nths thing [0..]}
\end{align*}
\]

*ISO> nth set 42
[1,3,5]

*ISO> take 3 (stream_of hfs)
[H [],H [H []],H [H [H []]]]

8.2 Random Instance Generation

Combining nth with a random generator for nat provides free algorithms for randomly generating complex objects of customizable size.

\[
\begin{align*}
\text{random_gen thing seed largest n} & = \text{genericTake n} \\
(\text{nths thing (rans seed largest)}) & \text{where} \\
\text{rans seed largest} & = \\
\text{randomRs (0,largest) (mkStdGen seed)}
\end{align*}
\]

*ISO> random_gen hfs 11 999 3
[[0,2,5],[0,5,9],[0,1,5,6]]

*ISO> head (random_gen hfs 7 30 1)
H [H [],H [H []],H [H [][]],H [H [H []]]]

This is useful for further automating test generators in tools like QuickCheck [9].

8.3 Succinct Representations

Depending on the information theoretical density of various data representations as well as on the constant factors involved in various data structures, significant data compression can be achieved by choosing an alternate isomorphic representation, as shown in the following examples.

*ISO> as hff hfs (H [H [H []],H [H [],
H [H [][]],H [H []],H [H [H [][]]])
H [H [H []],H [H []],H [H []]]

*ISO> as nat hff (H [H [H []],H [H []],H [H []]])
42

Sometimes, mapping to efficient arbitrary length integer implementations (usually C-based libraries) can provide more compact representations or improved performance for isomorphic higher level data representations, such as HFF or HFS.
In other cases, like $H [ H [... [ ]... ] ]$ corresponding to a tower of exponents as a hereditarily finite function or set, such representations are more compact than integers or list of integers representations that would quickly overflow all available memory.

### 8.4 Experimental Mathematics

We can compare representations sharing a common data type to conjecture about their asymptotic information density.

For instance, after defining:

- \( \text{length\_as\ t} = \text{fit\ genericLength\ (with\ nat\ t)} \)
- \( \text{sum\_as\ t} = \text{fit\ sum\ (with\ nat\ t)} \)
- \( \text{size\_as\ t} = \text{fit\ tsize\ (with\ nat\ t)} \)

we can conjecture that finite functions are more compact than sets asymptotically:

*ISO> length\_as\ set 123456789012345678901234567890 54
*ISO> length\_as\ fun 123456789012345678901234567890 54
*ISO> sum\_as\ set 123456789012345678901234567890 2690
*ISO> sum\_as\ fun 123456789012345678901234567890 43

and then observe that the same trend applies also to their hereditarily finite derivatives:

*ISO> size\_as\ hfs 123456789012345678901234567890 627
*ISO> size\_as\ hff 123456789012345678901234567890 91

### 8.5 A Surprising Free Algorithm: strange\_sort

A simple isomorphism like nat\_set can exhibit interesting properties as a building block of more intricate mappings, such as Ackermann’s encoding. Here is another simple free algorithm for sorting a list of distinct elements without the explicit use of comparison operations.

strange\_sort :: [Nat] \rightarrow [Nat]
strange\_sort = (from nat\_set). (to nat\_set)

*ISO> strange\_sort [2,9,3,1,5,0,7,4,8,6]
[0,1,2,3,4,5,6,7,8,9]

This algorithm emerges as a consequence of the commutativity of addition and the unicity of the decomposition of a natural number as a sum of powers of 2. Note that this property is derived from the particular choice of the isomorphism between natural numbers and sets.
### 8.6 Other Applications

A fairly large number of useful algorithms in fields ranging from data compression, coding theory, and cryptography to compilers, circuit design, and computational complexity involve bijective functions between heterogeneous data types. Their systematic encapsulation in a generic API that coexists well with strong typing can bring significant simplifications to various software modules with the added benefits of reliability and easier maintenance. In a genetic programming context [13], the use of isomorphisms between bitvectors/natural numbers on one side, and trees/graphs representing HFSs, HFFs on the other side, looks like a promising phenotype-genotype connection. Mutations and crossovers in a data type close to the problem domain are transparently mapped to numerical domains where evaluation functions can be computed easily. In the context of software transaction memory implementations (e.g., Haskell’s STM [14]), encodings through isomorphisms are subject to efficient shortcuts; for example, undo operations in the case of transaction failures can be performed by applying inverse transformations without needing to save the intermediate chain of data structures involved.

### 9. Related Work

The closest reference on encapsulating bijections as a data type is [15] and Conal Elliott’s composable bijections Haskell module [16], where, in a more complex setting, Arrows [17] are used as the underlying abstractions. While our Iso data type is similar to the Bij data type in [16] and BiArrow concept of [15], the techniques for using such isomorphisms as building blocks of an embedded composition language centered around encodings as natural numbers are new.

Ranking functions can be traced back to Gödel numberings [7, 8] associated to formulas. Together with their inverse unranking functions they are also used in combinatorial generation algorithms [18–21]. However, the generic view of such transformations as hylo-morphisms obtained compositionally from simpler isomorphisms, as described in this paper, is new.

Natural number encodings of hereditarily finite sets have triggered the interest of researchers in fields ranging from axiomatic set theory and foundations of logic to complexity theory and combinatorics [22–27]. Computational and data representation aspects of finite set theory have been described in logic programming and theorem proving contexts in [28, 29].

Pairing functions have been used in work on decision problems as early as [30, 31]. A typical use in the foundations of mathematics is [32]. An extensive study of various pairing functions and their computational properties is presented in [33].
10. Conclusion

We have shown the expressiveness of Haskell as a metalanguage for executable mathematics by describing encodings for functions and finite sets in a uniform framework as data type isomorphisms with a groupoid structure. Haskell’s higher order functions and recursion patterns have helped the design of an embedded data transformation language. The framework has been extended with hylomorphisms providing generic mechanisms for encoding hereditarily finite sets and hereditarily finite functions. In the process, a few surprising free algorithms have emerged, including Ackermann’s encoding from hereditarily finite sets to natural numbers. A longer version of this paper, covering isomorphisms to data types as parenthesis languages, dyadic rationals, functional binary numbers, permutations, binary decision diagrams, and their hereditarily finite derivatives is available at arXiv.org/abs/0808.2953. A Mathematica notebook, combining a superset of the code in the paper with interactive visualizations of various transformations, is available at http://logic.cse.unt.edu/tarau/research/2009/isoNKS.nb.

References


Complex Systems, 18 © 2010 Complex Systems Publications, Inc.
https://doi.org/10.25088/ComplexSystems.18.4.475


